NAG Fortran Library Routine Document

E04NDF/E04NDA

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

To supply optional parameters to E04NCF/E04NCA from an external file. More precisely, E04NDF must be used to supply optional parameters to E04NCF and E04NDA must be used to supply optional parameters to E04NCA.

E04NDA is a version of E04NDF that has additional parameters in order to make it safe for use in multithreaded applications (see Section 5 below). The initialisation routine E04WBF **must** have been called prior to calling E04NDA.

2 Specifications

2.1 Specification for E04NDF

```
SUBROUTINE E04NDF(IOPTNS, INFORM)
INTEGER           IOPTNS, INFORM
```

2.2 Specification for E04NDA

```
SUBROUTINE E04NDA(IOPTNS, LWSAV, IWSAV, RWSAV, INFORM)
INTEGER           IOPTNS, IWSAV(610), INFORM
real              RWSAV(475)
LOGICAL           LWSAV(120)
```

3 Description

E04NDF/E04NDA may be used to supply values for optional parameters to the corresponding routines E04NCF/E04NCA. E04NDF/E04NDA reads an external file and each line of the file defines a single optional parameter. It is only necessary to supply values for those parameters whose values are to be different from their default values.

Each optional parameter is defined by a single character string, of up to 72 characters, consisting of one or more items. The items associated with a given option must be separated by spaces, or equals signs [=]. Alphabetic characters may be upper or lower case. The string

```
Print level = 1
```

is an example of a string used to set an optional parameter. For each option the string contains one or more of the following items:

(a) A mandatory keyword.

(b) A phrase that qualifies the keyword.

(c) A number that specifies an INTEGER or ***real*** value. Such numbers may be up to 16 contiguous characters in Fortran's I, F, E or D formats, terminated by a space if this is not the last item on the line.

Blank strings and comments are ignored. A comment begins with an asterisk (*) and all subsequent characters in the string are regarded as part of the comment.

The file containing the options must start with **begin** and must finish with **end**. An example of a valid options file is:

```
Begin * Example options file
   Print level = 5
End
```

For E04NDF each line of the file is normally printed as it is read, on the current advisory message unit (see X04ABF), but printing may be suppressed using the keyword **nolist**. To suppress printing of **begin**, **nolist** must be the first option supplied as in the file:

```
Begin
   Nolist
   Print level = 5
End
```

Printing will automatically be turned on again after a call to E04NCF or E04NDF and may be turned on again at any time using the keyword **list**.

For E04NDA printing is turned off by default, but may be turned on at any time using the keyword **list**.

Optional parameter settings are preserved following a call to E04NCF/E04NCA and so the keyword **defaults** is provided to allow you to reset all the optional parameters to their default values prior to a subsequent call to E04NCF/E04NCA.

A complete list of optional parameters, their abbreviations, synonyms and default values is given in Section 11 of the document for E04NCF/E04NCA.

4 References

None.

5 Parameters

1: IOPTNS – INTEGER *Input*

On entry: the unit number of the options file to be read.

Constraint: $0 \leq \text{IOPTNS} \leq 99$.

2: INFORM – INTEGER *Output*

Note: *for E04NDA, INFORM does not occur in this position in the parameter list. See the additional parameters described below.*

On exit: contains zero if the options file has been successfully read and a value > 0 otherwise (see Section 6).

Note: *the following are additional parameters for specific use with E04NDA. Users of E04NDF therefore need not read the remainder of this section.*

2: LWSAV(120) – LOGICAL array *Workspace*

3: IWSAV(610) – INTEGER array *Workspace*

4: RWSAV(475) – ***real*** array *Workspace*

The arrays LWSAV, IWSAV and RWSAV **must not** be altered between calls to any of the routines E04WBF, E04NCA, E04NDA or E04NEA.

5: INFORM – INTEGER *Output*

On exit: contains zero if the options file has been successfully read and a value > 0 otherwise (see Section 6).

6 Error Indicators and Warnings

Errors or warnings detected by the routine:

INFORM = 1

IOPTNS is not in the range $[0, 99]$.

INFORM = 2

begin was found, but end-of-file was found before **end** was found.

INFORM = 3

end-of-file was found before **begin** was found.

INFORM = 4

Not used.

INFORM = 5

One or more lines of the options file is invalid. Check that all keywords are neither ambiguous nor misspelt.

7 Accuracy

Not applicable.

8 Further Comments

E04NEF/E04NEA may also be used to supply optional parameters to the corresponding routines E04NCF/E04NCA.

9 Example

To minimize the quadratic function $c^T x + \frac{1}{2} x^T A x$, where

$$c = (-4.0,\ -1.0,\ -1.0,\ -1.0,\ -1.0,\ -1.0,\ -1.0,\ -0.1,\ -0.3)^T,$$

$$A = \begin{pmatrix} 2 & 1 & 1 & 1 & 1 & 0 & 0 & 0 & 0 \\ 1 & 2 & 1 & 1 & 1 & 0 & 0 & 0 & 0 \\ 1 & 1 & 2 & 1 & 1 & 0 & 0 & 0 & 0 \\ 1 & 1 & 1 & 2 & 1 & 0 & 0 & 0 & 0 \\ 1 & 1 & 1 & 1 & 2 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \end{pmatrix}$$

subject to the bounds

$$\begin{aligned} -2 \leq x_1 \leq 2 \\ -2 \leq x_2 \leq 2 \\ -2 \leq x_3 \leq 2 \\ -2 \leq x_4 \leq 2 \\ -2 \leq x_5 \leq 2 \\ -2 \leq x_6 \leq 2 \\ -2 \leq x_7 \leq 2 \\ -2 \leq x_8 \leq 2 \\ -2 \leq x_9 \leq 2 \end{aligned}$$

and to the general constraints

$$\begin{aligned} -2.0 &\leq x_1 + x_2 + x_3 + x_4 + x_5 + x_6 + x_7 + x_8 + 4x_9 \leq 1.5 \\ -2.0 &\leq x_1 + 2x_2 + 3x_3 + 4x_4 - 2x_5 + x_6 + x_7 + x_8 + x_9 \leq 1.5 \\ -2.0 &\leq x_1 - x_2 + x_3 - x_4 + x_5 + x_6 + x_7 + x_8 + x_9 \leq 4.0 \end{aligned}$$

The initial point, which is feasible, is

$$x_0 = (0,\ 0,\ 0,\ 0,\ 0,\ 0,\ 0,\ 0,\ 0)^T,$$

and $F(x_0) = 0$.

The optimal solution (to five figures) is

$$x^* = (2.0, -0.23333, -0.26667, -0.3, -0.1, 2.0, 2.0, -1.7777, -0.45555)^T,$$

and $F(x^*) = -8.0678$. Three bound constraints and two general constraints are active at the solution. Note that, although the Hessian matrix is positive semi-definite, the point x^* is unique.

In this example the options file read by E04NDF/E04NDA is appended to the data file for the program (see Section 9.2). It would usually be more convenient in practice to keep the data file and the options file separate.

9.1 Program Text

Note: the listing of the example program presented below uses ***bold italicised*** terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

Note: *the following program illustrates the use of E04NDF. An equivalent program illustrating the use of E04NDA is available with the supplied Library and is also available from the NAG web site.*

```
*     E04NDF Example Program Text
*     Mark 16 Release. NAG Copyright 1993.
*     .. Parameters ..
      INTEGER          NIN, NOUT
      PARAMETER        (NIN=5,NOUT=6)
      INTEGER          MMAX, NMAX, NCMAX
      PARAMETER        (MMAX=10,NMAX=10,NCMAX=10)
      INTEGER          LDC, LDA
      PARAMETER        (LDC=NCMAX,LDA=MMAX)
      INTEGER          LIWORK, LWORK
      PARAMETER        (LIWORK=100,LWORK=1000)
*     .. Local Scalars ..
      real             OBJ
      INTEGER          I, IFAIL, INFORM, ITER, J, M, N, NCLIN
*     .. Local Arrays ..
      real             A(LDA,NMAX), B(MMAX), BL(NMAX+NCMAX),
     +                 BU(NMAX+NCMAX), C(LDC,NMAX), CLAMDA(NMAX+NCMAX),
     +                 CVEC(NMAX), WORK(LWORK), X(NMAX)
      INTEGER          ISTATE(NMAX+NCMAX), IWORK(LIWORK), KX(NMAX)
*     .. External Subroutines ..
      EXTERNAL         E04NCF, E04NDF, E04NEF, X04ABF
*     .. Executable Statements ..
      WRITE (NOUT,*) 'E04NDF Example Program Results'
*     Skip heading in data file
      READ (NIN,*)
      READ (NIN,*) M, N, NCLIN
      IF (M.LE.MMAX .AND. N.LE.NMAX .AND. NCLIN.LE.NCMAX) THEN
*
*        Read CVEC, A, C, BL, BU and X from data file
*
         READ (NIN,*) (CVEC(I),I=1,N)
         READ (NIN,*) ((A(I,J),J=1,N),I=1,M)
         READ (NIN,*) ((C(I,J),J=1,N),I=1,NCLIN)
         READ (NIN,*) (BL(I),I=1,N+NCLIN)
         READ (NIN,*)
         READ (NIN,*) (BU(I),I=1,N+NCLIN)
         READ (NIN,*)
         READ (NIN,*) (X(I),I=1,N)
```

```
*
*        Set two options using E04NEF
*
         CALL E04NEF(' Infinite Bound Size = 1.0D+25 ')
*
         CALL E04NEF(' Problem Type = QP2 ')
*
*        Set the unit number for advisory messages to NOUT
*
         CALL X04ABF(1,NOUT)
*
*        Read the options file for the remaining options
*
         CALL E04NDF(NIN,INFORM)
*
         IF (INFORM.NE.0) THEN
            WRITE (NOUT,99999) 'E04NDF terminated with INFORM = ',
     +        INFORM
            STOP
         END IF
*
*        Solve the problem
*
         IFAIL = -1
*
         CALL E04NCF(M,N,NCLIN,LDC,LDA,C,BL,BU,CVEC,ISTATE,KX,X,A,B,
     +              ITER,OBJ,CLAMDA,IWORK,LIWORK,WORK,LWORK,IFAIL)
*
      END IF
      STOP
*
99999 FORMAT (1X,A,I3)
      END
```

9.2 Program Data

```
E04NDF Example Program Data
  9   9   3                                    :Values of M, N and NCLIN
 -4.0  -1.0  -1.0  -1.0  -1.0  -1.0  -1.0  -0.1  -0.3        :End of CVEC
  2.0   1.0   1.0   1.0   1.0   0.0   0.0   0.0   0.0
  1.0   2.0   1.0   1.0   1.0   0.0   0.0   0.0   0.0
  1.0   1.0   2.0   1.0   1.0   0.0   0.0   0.0   0.0
  1.0   1.0   1.0   2.0   1.0   0.0   0.0   0.0   0.0
  1.0   1.0   1.0   1.0   2.0   0.0   0.0   0.0   0.0
  0.0   0.0   0.0   0.0   0.0   0.0   0.0   0.0   0.0
  0.0   0.0   0.0   0.0   0.0   0.0   0.0   0.0   0.0
  0.0   0.0   0.0   0.0   0.0   0.0   0.0   0.0   0.0
  0.0   0.0   0.0   0.0   0.0   0.0   0.0   0.0   0.0        :End of matrix A
  1.0   1.0   1.0   1.0   1.0   1.0   1.0   1.0   4.0
  1.0   2.0   3.0   4.0  -2.0   1.0   1.0   1.0   1.0
  1.0  -1.0   1.0  -1.0   1.0   1.0   1.0   1.0   1.0        :End of matrix C
 -2.0  -2.0  -2.0  -2.0  -2.0  -2.0  -2.0  -2.0  -2.0  -2.0  -2.0  -2.0
                                                            :End of BL
  2.0   2.0   2.0   2.0   2.0   2.0   2.0   2.0   2.0   1.5   1.5   4.0
                                                            :End of BU
  0.0   0.0   0.0   0.0   0.0   0.0   0.0   0.0   0.0        :End of X
Begin   Example options file for  E04NDF
   Iteration Limit   =  30      *  (Default =  90)
End
```

9.3 Program Results

```
 E04NDF Example Program Results


 Calls to E04NEF
 ---------------

       Infinite Bound Size = 1.0E+25
       Problem Type = QP2


 OPTIONS file
 ------------

      Begin   Example options file for  E04NDF
         Iteration Limit   =  30      *  (Default =  90)
      End

 *** E04NCF
 *** Start of NAG Library implementation details ***

 Implementation title: Generalised Base Version
            Precision: FORTRAN double precision
         Product Code: FLBAS20D
                 Mark: 20A

 *** End of NAG Library implementation details ***

 Parameters
 ----------

 Problem type...........       QP2       Hessian................        NO

 Linear constraints.....         3       Feasibility tolerance..  1.05E-08
 Variables..............         9       Crash tolerance........  1.00E-02
 Objective matrix rows..         9       Rank tolerance.........  1.05E-07

 Infinite bound size....  1.00E+25       COLD start.............
 Infinite step size.....  1.00E+25       EPS (machine precision)  1.11E-16

 Print level............        10       Feasibility phase itns.        60
 Monitoring file........        -1       Optimality  phase itns.        30

 Workspace provided is     IWORK(     100),  WORK(    1000).
 To solve problem we need  IWORK(       9),  WORK(     270).

 Rank of the objective function data matrix =     5


 Itn     Step Ninf Sinf/Objective  Norm Gz
   0  0.0E+00    0   0.000000E+00  4.5E+00
   1  7.5E-01    0  -4.375000E+00  5.0E-01
   2  1.0E+00    0  -4.400000E+00  2.8E-17
   3  3.0E-01    0  -4.700000E+00  8.9E-01
   4  1.0E+00    0  -5.100000E+00  3.4E-17
   5  5.4E-01    0  -6.055714E+00  1.7E+00
   6  1.1E-02    0  -6.113326E+00  1.6E+00
   7  1.1E-01    0  -6.215049E+00  1.2E+00
   8  1.0E+00    0  -6.538008E+00  3.5E-17
   9  6.5E-01    0  -7.428704E+00  7.2E-02
  10  1.0E+00    0  -7.429717E+00  1.8E-17
  11  1.0E+00    0  -8.067718E+00  5.6E-17
  12  1.0E+00    0  -8.067778E+00  5.6E-17

 Exit from QP problem after    12 iterations.


 Varbl State     Value       Lower Bound   Upper Bound    Lagr Mult   Slack

 V   1    UL    2.00000      -2.00000       2.00000     -0.8000         .
```

```
 V   2    FR  -0.233333      -2.00000       2.00000           .       1.767
 V   3    FR  -0.266667      -2.00000       2.00000           .       1.733
 V   4    FR  -0.300000      -2.00000       2.00000           .       1.700
 V   5    FR  -0.100000      -2.00000       2.00000           .       1.900
 V   6    UL    2.00000      -2.00000       2.00000     -0.9000         .
 V   7    UL    2.00000      -2.00000       2.00000     -0.9000         .
 V   8    FR   -1.77778      -2.00000       2.00000           .      0.2222
 V   9    FR  -0.455556      -2.00000       2.00000           .       1.544

 L Con State     Value       Lower Bound   Upper Bound    Lagr Mult   Slack

 L   1    UL    1.50000      -2.00000       1.50000     -6.6667E-02  6.6613E-16
 L   2    UL    1.50000      -2.00000       1.50000     -3.3333E-02  8.8818E-16
 L   3    FR    3.93333      -2.00000       4.00000           .      6.6667E-02

 Exit E04NCF - Optimal QP solution.

 Final QP objective value =   -8.067778
```